```
// Theorem 1
F := Rationals();
F<d,t,h0,h1,e> := FunctionField(F, 5);
// Here, h0, h1 are the canonical heights rather than the naive ones
D := t^2 - 4*d;
detM2 := 4*h0*h1 - e^2;

PR<v1> := PolynomialRing(F);
v0 := t*e - v1;
M4 := Matrix(PR, 4, 4, [
    2*h0, t*h0,   e,    v0,
    t*h0, 2*d*h0, v1,   d*e,
    e,    v1,     2*h1, t*h1,
    v0,   d*e,    t*h1, 2*d*h1
]);

detM4 := Determinant(M4);
rho := v1^2 - t*e*v1 + (d*e^2 + D*h0*h1);
detM4 eq rho^2;
Discriminant(rho) eq -D*detM2;
```

```
////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////////
```

```
// Mestre-Kuwata-Wang's formulas
F := Rationals();
F<a,b> := FunctionField(F, 2);
F<u> := FunctionField(F);

x0 := b*(u^6 - 1) / (a*u^2*(1 - u^4));
x1 := x0*u^2;
y := u^3;
fx0 := x0^3 + a*x0 + b;
fx1 := x1^3 + a*x1 + b;
fx0*y^2 eq fx1;

g := fx0;
num := b*(b^2*u^12 + 3*b^2*u^10 + (a^3 + 6*b^2)*u^8 +
       (2*a^3 + 7*b^2)*u^6 + (a^3 + 6*b^2)*u^4 + 3*b^2*u^2 + b^2);
den := -(a*u^2*(u^2 + 1))^3;
g eq num/den;

P2<X,Y,Z> := ProjectivePlane(F);
f := X^3 + a*X*Z^2 + b*Z^3;
Eg := Curve(P2, g*Y^2*Z - f);
```

```
P0 := Eg ! [x0, 1, 1];
P1 := Eg ! [x1, y, 1];
O := Eg ! [0, 1, 0];
W, EgToW := EllipticCurve(Eg, O);

P0 := EgToW(P0);
P1 := EgToW(P1);
Height(P0);
Height(P1);
HeightPairing(P0, P1);
```